Introduction to PIC Programming

Baseline Architecture and Assembly Language

by David Meiklejohn, Gooligum Electronics

Lesson 8: Driving 7-Segment Displays

The applications we've looked at so far have used only one or two LEDs as outputs. That's enough for simple indicators, but many applications need to be able to display information in numeric, alphanumeric or graphical form. Although LCD and OLED displays are becoming more common, there is still a place, when displaying numeric (or sometimes hexadecimal) information, for 7-segment LED displays.

To drive a single 7-segment display, in a straightforward manner, we need seven outputs. That rules out the PIC12F509 we've been using so far. Its bigger brother, the 14-pin 16F505, is quite suitable, but to avoid using too many different devices, we'll jump to the more capable 16F506. In fact, the 16F506 can be made to drive up to four 7-segment displays, using a technique known as *multiplexing*. But to display even a single digit, that digit has to be translated into a specific pattern of segments in the display. That translation is normally done through *lookup tables*.

In summary, this lesson covers:

- Introductory overview of the PIC16F506 MCU
- Driving a single 7-segment display
- Using lookup tables
- Using multiplexing to drive multiple displays
- Binary-coded decimal (BCD)

Introducing the PIC16F506

The previous lessons have focussed on the 10F200 (or 12F508) and 12F509.

We saw in lesson 1 that the 12F508 and 12F509 are part of a family which includes the 14-pin 16F505. That lesson included the following table, summarising the differences within the 12F508/12F509/16F505 family:

Device	Program Memory (words)	Data Memory (bytes)	Package	I/O pins	Clock rate (maximum)
12F508	512	25	8-pin	6	4 MHz
12F509	1024	41	8-pin	6	4 MHz
16F505	1024	72	14-pin	12	20 MHz

Although the 16F505 is architecturally very similar to the 12F508/509, it has more data memory, more I/O pins (11 I/O and 1 input-only), a higher maximum clock speed and wider range of oscillator options.

The 12F510 and 16F506 form a very similar family, adding peripherals with *analog* (continuously variable) inputs: analog comparators and an analog-to-digital converter (ADC). We'll explore those capabilities in

lessons 9 and 10, but briefly – a comparator allows us to compare two analog signals (one of which is often a fixed reference voltage), while the ADC allows us to measure analog signals.

The following table compares the features of the devices in both families:

Device	Program Memory (words)	Data Memory (bytes)	Package	I/O pins	Comp-arators	Analog Inputs	Clock rate (maximum)
12F508	512	25	8-pin	6	-	-	4 MHz
12F509	1024	41	8-pin	6	-	-	4 MHz
12F510	1024	38	8-pin	6	1	3	8 MHz
16F505	1024	72	14-pin	12	-	-	20 MHz
16F506	1024	67	14-pin	12	2	3	20 MHz

Although the 16F505 would be adequate for this lesson, we may as well jump directly to the 16F506, which does everything the 16F505 does (although it does have 5 bytes less data memory…) and continue to use it when we look at analog inputs in the upcoming lessons. We'll just ignore the analog side for now.

The expanded capabilities of the 16F506 (other than analog) are detailed in the following sections.

Additional oscillator options

The 16F506 supports an expanded range of oscillator options, selected by bits in the configuration word:

Bit 11	Bit 10	Bit 9	Bit 8	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
-	-	-	-	-	IOSCFS	MCLRE	$\overline{CP}$	WDTE	FOSC2	FOSC1	FOSC0

In the 12F510/16F506 devices, the internal RC oscillator can optionally run at a nominal 8 MHz instead of 4 MHz. *Be careful, if you select 8 MHz, that any code (such as delays) written for a 4 MHz clock is correct.*

The speed of the internal RC oscillator is selected by the IOSCFS bit.

Setting IOSCFS to '1' (by ANDing the symbol '`_IOSCFS_ON`' into the configuration word expression) selects 8 MHz operation; clearing it to '0' (with '`_IOSCFS_OFF`') selects 4 MHz.

The three FOSC bits allow the selection of eight clock options (twice the number available in the 12F509), as in the table below.

FOSC<2:0>	Standard symbol	Oscillator configuration
000	`_LP_OSC`	LP oscillator
001	`_XT_OSC`	XT oscillator
010	`_HS_OSC`	HS oscillator
011	`_EC_RB4EN`	EC oscillator + RB4
100	`_IntRC_OSC_RB4EN`	Internal RC oscillator + RB4
101	`_IntRC_OSC_CLKOUTEN`	Internal RC oscillator + CLKOUT
110	`_ExtRC_OSC_RB4EN`	External RC oscillator + RB4
111	`_ExtRC_OSC_CLKOUTEN`	External RC oscillator + CLKOUT

The 'LP' and 'XT' oscillator options are exactly the same as described in lesson 7: 'LP' mode being typically used to drive crystals with a frequency less than 200 kHz, and 'XT' mode being intended for crystals or resonators with a frequency between 200 kHz and 4 MHz.

The 'HS' ("high speed") mode extends this to 20 MHz. The crystal or resonator, with appropriate loading capacitors, is connected between the OSC1 and OSC2 pins in exactly the same way as for the 'LP' or 'XT' modes.

As explained in lesson 7, the 'LP' and 'XT' (and indeed 'HS') modes can be used with an external clock signal, driving the OSC1, or CLKIN, pin. The downside to using the "crystal" modes with an external clock is that the OSC2 pin remains unused, wasting a potentially valuable I/O pin.

The 'EC' oscillator mode addresses this problem. It is designed for use with an external clock signal driving the CLKIN pin, the same as is possible in the crystal modes, but with the significant advantage that the "OSC2 pin", pin 3 on the 16F506, is available for digital I/O as pin 'RB4'.

There are now two internal RC oscillator modes. '`_IntRC_OSC_RB4EN`' is just like the 12F509's '`_IntRC_OSC`' mode, where the internal RC oscillator runs (at either 4 MHz or 8 MHz on the 16F506) leaving all pins available for digital I/O – including RB4 (pin 3).

The second internal RC option, '`_IntRC_OSC_CLKOUTEN`', assigns pin 3 as 'CLKOUT' instead of RB4. In this mode, the instruction clock, which runs at one quarter the speed of the processor clock, i.e. a nominal 1 MHz (or 2 MHz if IOSCFS is set), is output on the CLKOUT pin. This output clock signal can be used to provide a clock signal to external devices, or for synchronising other devices with the PIC.

Lesson 7 showed how an external RC oscillator can be used with the 12F509. Although this mode usefully allows for low cost, low power operation, it has the same drawback as the externally-clocked "crystal" modes: pin 3 (OSC2) cannot be used for anything.

The external RC oscillator modes on the 16F506 overcome this drawback. In the first option, '`_ExtRC_OSC_RB4EN`', pin 3 is available for digital I/O as RB4.

The other external RC option, '`_ExtRC_OSC_CLKOUTEN`', assigns pin 3 to CLKOUT, with the instruction clock appearing as an output signal, running at one quarter the rate of the external RC oscillator (FOSC/4).

In summary, the expanded range of clock options provides for higher speed operation, more usable I/O pins, or a clock output to allow for external device synchronisation.

Additional I/O pins

The 16F506 provides twelve I/O pins (one being input-only), compared with the six (with one being input-only) available on the 12F508/509/510.

Twelve is too many pins to represent in a single 8-bit register, so instead of a single port named GPIO, the 16F506 has two ports, named PORTB and PORTC.

Six I/O pins are allocated to each port:

	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
PORTB			RB5	RB4	RB3	RB2	RB1	RB0
PORTC			RC5	RC4	RC3	RC2	RC1	RC0

The direction of each I/O pin is controlled by corresponding TRIS registers:

	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
TRISB			RB5	RB4		RB2	RB1	RB0
TRISC			RC5	RC4	RC3	RC2	RC1	RC0

As in the 12F509, the TRIS registers are not mapped into data memory and can only be accessed through the '`tris`' instruction, with an operand of 6 (or '`PORTB`') to load TRISB, or an operand of 7 (or '`PORTC`') to load TRISC.

RB3 is input only and, like GP3 on the 12F509, it shares a pin with $\overline{\text{MCLR}}$; the pin assignment being controlled by the MCLRE bit in the configuration word.

The 16F506 comes in a 14-pin package; the pin diagram is shown below.



Note that RC5 and T0CKI (the Timer0 external clock input) share the same pin.

We have seen that on the 12F509, T0CKI shares a pin with GP2, and to use GP2 as an output you must first disable T0CKI by clearing the T0CS bit in the OPTION register.

In the same way, to use RC5 as an output on the 16F506, you must first disable T0CKI by clearing T0CS.

The RB0, RB1 and RB2 pins are configured as analog inputs by default. To use any of these pins for digital I/O, they must be deselected as analog inputs. This can be done by clearing the ADCON0 register, as we'll see in lesson 10 on analog-to-digital conversion.

The RB0, RB1, RC0 and RC1 pins are configured as comparator inputs by default. To use any of these pins for digital I/O, the appropriate comparator must be disabled (by clearing the C1ON bit in the CM1CON0 register, and/or the C2ON bit in the CM2CON0 register), or its inputs reassigned, as explained in lesson 9.

> *Note: On PICs with comparators and/or analog (ADC) inputs, the comparator and analog inputs are enabled on start-up. To use a pin for digital I/O, any comparator or analog input assigned to that pin must first be disabled.*

This is a common trap for beginners, who wonder why their LED won't light, when they haven't deselected analog input on the pin they are using. That is why this tutorial series began with digital-only PICs.

For now, we'll just include the instructions to disable these analog inputs in the examples in this lesson, and leave the full explanations for lessons 9 and 10.

Additional data memory

The data memory, or register file, of the 16F506 is arranged in four banks, as follows:

PIC16F506 Registers

	Bank 0		*Bank 1*		*Bank 2*		*Bank 3*
00h	INDF	20h	INDF	40h	INDF	60h	INDF
01h	TMR0	21h	TMR0	41h	TMR0	61h	TMR0
02h	PCL	22h	PCL	42h	PCL	62h	PCL
03h	STATUS	23h	STATUS	43h	STATUS	63h	STATUS
04h	FSR	24h	FSR	44h	FSR	64h	FSR
05h	OSCCAL	25h	OSCCAL	45h	OSCCAL	65h	OSCCAL
06h	PORTB	26h	PORTB	46h	PORTB	66h	PORTB
07h	PORTC	27h	PORTC	47h	PORTC	67h	PORTC
08h	CM1CON0	28h	CM1CON0	48h	CM1CON0	68h	CM1CON0
09h	ADCON0	29h	ADCON0	49h	ADCON0	69h	ADCON0
0Ah	ADRES	2Ah	ADRES	4Ah	ADRES	6Ah	ADRES
0Bh	CM2CON0	2Bh	CM2CON0	4Bh	CM2CON0	6Bh	CM2CON0
0Ch	VRCON	2Ch	VRCON	4Ch	VRCON	6Ch	VRCON
0Dh – 0Fh	Shared GP Registers	2Dh – 2Fh	Map to Bank 0 0Dh – 0Fh	4Dh – 4Fh	Map to Bank 0 0Dh – 0Fh	6Dh – 6Fh	Map to Bank 0 0Dh – 0Fh
10h – 1Fh	General Purpose Registers	30h – 3Fh	General Purpose Registers	50h – 5Fh	General Purpose Registers	70h – 7Fh	General Purpose Registers

There are only 3 shared data registers (0Dh – 0Fh), which are mapped into all four banks.

In addition, there are 4 × 16 = 64 non-shared (*banked*) data registers, filling the top half of each bank.

Thus, the 16F506 has a total of 3 + 64 = 67 general purpose data registers.

The bank is selected by the FSR<6:5> bits, as was explained (for the 16F505) in lesson 3. Although an additional bank selection bit is used, compared with the single bit in the 12F509, you don't need to be aware of that; simply use the `banksel` directive in the usual way.

Driving a 7-segment LED Display

A 7-segment LED display is simply a collection of LEDs, typically one per segment (but often having two or more LEDs per segment for large displays), arranged in the "figure 8" pattern we are familiar with from numeric digital displays. 7-segment display modules also commonly include one or two LEDs for decimal points.

7-segment LED display modules come in one of two varieties: common-anode or common-cathode.

In a common-cathode module, the cathodes belonging to each segment are wired together within the module, and brought out through one or two (or sometimes more) pins. The anodes for each segment are brought out separately, each to its own pin. Typically, each segment (anode) would be connected to a separate output pin on the PIC, as shown in the following circuit diagram[1]:



The common cathode pins are connected together and grounded.

To light a given segment in a common-cathode display, the corresponding PIC output is set high. Current flows from the output and through the given segment (limited by a series resistor) to ground.

In a common-anode module, this is reversed; the anodes for each segment are wired together and the cathodes are accessible separately. In that case, the common anode pins are connected to the positive supply and each cathode is connected to a separate PIC output. To light a segment in a common-anode display, the corresponding PIC output is set low; current flows from the positive supply, through the segment and into the PIC's output.

Although, on the PIC16F506, a single pin can source or sink up to 25 mA, the maximum per port is 100 mA and the maximum current into V_{DD} (the device's supply current) is 150 mA. Given that the PIC itself consumes some current (up to around 2 mA) and that we'd potentially like to be able to draw current from the unused output pins, we should limit the total current drawn by the 7-segment display to no more than 100 mA or so. Since all the segments may be lit at once (when displaying '8'), we should to limit the current per pin to 100 mA ÷ 7 = 14.3 mA. The 330 Ω resistors limit the current to 10 mA, well within spec while giving a bright display.

If you are using the Gooligum baseline training board, you can implement this circuit by:

- placing shunts (six of them) across every position in jumper block JP4, connecting segments A-D, F and G to pins RB0-1 and RC1-4
- placing a single shunt in position 1 ("RA/RB4") of JP5, connecting segment E to pin RB4
- placing a shunt across pins 1 and 2 ("GND") of JP6, connecting digit 1 to ground.

All other shunts should be removed.

---

[1] The segment anodes are connected to PIC pins in the (apparently) haphazard way shown, because this reflects the connections on the Gooligum baseline training board. You'll often find that, by rearranging your PIC pin assignments, you can simplify your PCB layout and routing – even if it makes your schematic messier!

If you are using Microchip's Low Pin Count Demo Board, you will have to supply your own 7-segment display module, and connect it (and the current-limiting resistors) to the board. This can be done via the 14-pin header on the Low Pin Count Demo Board, as illustrated on the right. Note that the header pins corresponding to the "RB" pins on the 16F506 are labelled "RA" on the demo board, reflecting the PIC16F690 it is supplied with, not the 16F506 used here.



Be careful, because your 7-segment display module may have a different pin-out to that shown above. If you have a common-anode display, you will need to wire it correctly and make appropriate changes to the code presented here, but the techniques for driving the display are essentially the same.

Lookup tables

To display each digit, a corresponding pattern of segments must be lit, as follows:

Segment:	A	B	C	D	E	F	G
Pin:	RC1	RC2	RC4	RC3	RB4	RB1	RB0
0	on	on	on	on	on	on	off
1	off	on	on	off	off	off	off
2	on	on	off	on	on	off	on
3	on	on	on	on	off	off	on
4	off	on	on	off	off	on	on
5	on	off	on	on	off	on	on
6	on	off	on	on	on	on	on
7	on	on	on	off	off	off	off
8	on	on	on	on	on	on	on
9	on	on	on	on	off	on	on

We need a way to determine, or look up, the pattern corresponding to the digit to be displayed, and that is most effectively done with a *lookup table*.

The most common method of implementing lookup tables in the baseline PIC architecture is to use a *computed jump* into a table of '`retlw`' instructions.

For example, to look up the binary pattern to be applied to PORTC, corresponding to the digit in W, we could use the following subroutine:

```
; pattern table for 7 segment display on port C
;   RC4:1 = CDBA
get7sC  addwf   PCL,f
        retlw   b'011110'       ; 0
        retlw   b'010100'       ; 1
        retlw   b'001110'       ; 2
        retlw   b'011110'       ; 3
        retlw   b'010100'       ; 4
        retlw   b'011010'       ; 5
        retlw   b'011010'       ; 6
        retlw   b'010110'       ; 7
        retlw   b'011110'       ; 8
        retlw   b'011110'       ; 9
```

Baseline PICs have a single addition instruction: '`addwf f,d`' – "**add W** to **f**ile register", placing the result in the register if the destination is '`,f`', or in W if the destination is '`,w`'.

As mentioned in lesson 3, the program counter (PC) is a 12-bit register holding the full address of the next instruction to be executed. The lower eight bits of the program counter (PC<7:0>) are mapped into the PCL register. If you change the contents of PCL, you change the program counter – affecting which instruction will be executed next. For example, if you add 2 to PCL, the program counter will be advanced by 2, skipping the next two instructions.

In the code above, the first instruction adds the table index, or offset (corresponding to the digit being looked up), in W to PCL, writing the result back to PCL.

If W contains '0', 0 is added to PCL, leaving the program counter unchanged, and the next instruction is executed as normal: the first '`retlw`', returning the pattern for digit '0' in W.

But consider what happens if the subroutine is called with W containing '4'. PCL is incremented by 4, advancing the program counter by 4, so the next four instructions will be skipped. The fifth '`retlw`' instruction will be executed, returning the pattern for digit '4' in W.

This lookup table could then be used ('called', since it is actually a subroutine) as follows:

```
        movf    digit,w         ; get digit to display
        call    get7sC          ; lookup pattern for port C
        movwf   PORTC           ;   then output it
```

(assuming that the digit to be displayed is stored in a variable called '`digit`')

A second lookup table, called the same way, would be used to lookup the pattern to be output on PORTB.

The define table directive

Since lookup tables are very useful, and commonly used, the MPASM assembler provides a shorthand way to define them: the '`dt`' (short for "define table") directive. Its syntax is:

```
[label] dt      expr1[,expr2,…,exprN]
```

where each expression is an 8-bit value. This generates a series of `retlw` instructions, one for each expression. The directive is equivalent to:

```
[label] retlw   expr1
        retlw   expr2
        …
        retlw   exprN
```

Thus, we could write the code above as:

```
get7sC  addwf   PCL,f
        dt      b'011110',b'010100',b'001110',b'011110',b'010100'  ; 0,1,2,3,4
        dt      b'011010',b'011010',b'010110',b'011110',b'011110'  ; 5,6,7,8,9
```

or it could even be written as:

```
get7sC  addwf   PCL,f
        dt      0x1E,0x14,0x0E,0x1E,0x14,0x1A,0x1A,0x16,0x1E,0x1E   ; 0-9
```

Of course, the `dt` directive is more appropriate in some circumstances than others. Your table may be easier to understand if you use only one expression per line, in which case it is clearer to simply use `retlw`.

A special case where '`dt`' makes your code much more readable is with text strings. For example:

```
        dt      "Hello world",0
```

is equivalent to:

```
        retlw   'H'
        retlw   'e'
        retlw   'l'
        retlw   'l'
        retlw   'o'
        retlw   ' '
        retlw   'w'
        retlw   'o'
        retlw   'r'
        retlw   'l'
        retlw   'd'
        retlw   0
```

The '`dt`' form is clearly preferable in this case.

Lookup table address limitation

A significant limitation of the baseline PIC architecture is that, when any instruction modifies PCL, bit 8 of the program counter (PC<8>) is cleared. That means that, whatever the result of the table offset addition, when PCL is updated, the program counter will be left pointing at an address in the first 256 words of the current program memory page (PC<9> is updated from the PA0 bit, in the same way as for a `goto` or `call` instruction; see lesson 3.)

This is very similar to the address limitation, discussed in lesson 3, which applies to subroutines on baseline PICs. But the constraint on lookup tables is even more limiting – because the result of the offset addition must be within the first 256 words of a page, not just the start of the table, the whole table has to fit within the first 256 words of a page.

> *In the baseline PIC architecture,* ***lookup tables must be wholly contained within the first 256 locations of a program memory page.***

We have seen that a workaround for the limitation on subroutine addressing is to use a vector table, but no such workaround is possible for lookup tables.

Therefore you must take care to ensure that any lookup tables are located toward the beginning of a program memory page. A simple way to do that is to place the lookup tables in a separate code section, located explicitly at the start of a page, by specifying its address with the `CODE` directive.

For example:

```
TABLES  CODE    0x200           ; locate at beginning of a page

; pattern table for 7 segment display on port B
;   RB4 = E, RB1:0 = FG
get7sB  addwf   PCL,f
        retlw   b'010010'       ; 0
        retlw   b'000000'       ; 1
        retlw   b'010001'       ; 2
        retlw   b'000001'       ; 3
        retlw   b'000011'       ; 4
        retlw   b'000011'       ; 5
        retlw   b'010011'       ; 6
        retlw   b'000000'       ; 7
        retlw   b'010011'       ; 8
        retlw   b'000011'       ; 9
```

This places the tables explicitly at the beginning of page 1 (the 16F506 has two program memory pages), out of the way of the start-up code located at the beginning of page 0 (0x000).

This means of course that you need to use the `pagesel` directive if calling these lookup tables from a different code section.

To display a digit, we need to look up and then write the correct patterns for ports B and C, meaning two table lookups for each digit displayed.

Ideally we'd have a single routine which, given the digit to be displayed, performs the table lookups and writes the patterns to the I/O ports. To avoid the need for multiple `pagesel` directives, this "display digit" subroutine can be located on the same page as the lookup tables.

For example:

```
TABLES  CODE    0x200           ; locate at beginning of a page

; pattern table for 7 segment display on port B
;   RB4 = E, RB1:0 = FG
get7sB  addwf   PCL,f
        retlw   b'010010'       ; 0
        retlw   b'000000'       ; 1
        ... (etc.)

; pattern table for 7 segment display on port C
;   RC4:1 = CDBA
get7sC  addwf   PCL,f
        retlw   b'011110'       ; 0
        retlw   b'010100'       ; 1
        ... (etc.)

; Display digit passed in 'digit' variable on 7-segment display
set7seg_R
        movf    digit,w         ; get digit to display
        call    get7sB          ; lookup pattern for port B
        movwf   PORTB           ;   then output it
        movf    digit,w         ; repeat for port C
        call    get7sC
        movwf   PORTC
        retlw   0
```

Then to display a digit, it is simply a matter of writing the value into the '`digit`' variable (assumed to be in a shared data segment to avoid the need for banking), and calling the '`set7seg_R`' routine.

Note that it's assumed that the '`set7seg_R`' routine is called through a vector in page 0 labelled '`set7seg`', so that the subroutine doesn't have to be in the first 256 words of page 1; it can be anywhere on page 1 and we still avoid the need for a '`pagesel`' when calling the lookup tables from it.

So, given these lookup tables and a subroutine that will display a selected digit, what to do with them? We've been blinking LEDs at 1 Hz, so counting seconds seems appropriate.

Complete program

The following program incorporates the code fragments presented above, and code (e.g. macros) and techniques from previous lessons, to count repeatedly from 0 to 9, with 1 s between each count.

```
;*****************************************************************
;   Description:    Lesson 8, example 1a                        *
;                                                               *
;   Demonstrates use of lookup tables to drive 7-segment display*
;                                                               *
;   Single digit 7-segment LED display counts repeating 0 -> 9  *
;   1 second per count, with timing derived from int 4 MHz oscillator  *
;                                                               *
;*****************************************************************
;                                                               *
;   Pin assignments:                                            *
;       RB0-1,RB4, RC1-4 = 7-segment display bus (common cathode)   *
;                                                               *
;*****************************************************************

    list        p=16F506
    #include    <p16F506.inc>

    #include    <stdmacros-base.inc>    ; DelayMS - delay in milliseconds
                                        ;   (calls delay10)
    EXTERN      delay10_R               ; W x 10 ms delay

    radix       dec


;***** CONFIGURATION
                ; ext reset, no code protect, no watchdog, 4 MHz int clock
    __CONFIG    _MCLRE_ON & _CP_OFF & _WDT_OFF & _IOSCFS_OFF & _IntRC_OSC_RB4EN


;***** VARIABLE DEFINITIONS
        UDATA_SHR
digit   res 1                   ; digit to be displayed


;***** RC CALIBRATION
RCCAL   CODE    0x3FF           ; processor reset vector
        res 1                   ; holds internal RC cal value, as a movlw k


;***** RESET VECTOR ****************************************************
RESET   CODE    0x000           ; effective reset vector
        movwf   OSCCAL          ; apply internal RC factory calibration
        pagesel start
        goto    start           ; jump to main code
```

```
;***** Subroutine vectors
delay10                         ; delay W x 10 ms
        pagesel delay10_R
        goto    delay10_R
set7seg                         ; display digit on 7-segment display
        pagesel set7seg_R
        goto    set7seg_R

;***** MAIN PROGRAM ****************************************************
MAIN    CODE

;***** Initialisation
start
        ; configure ports
        clrw                    ; configure PORTB and PORTC as all outputs
        tris    PORTB
        tris    PORTC
        clrf    ADCON0          ; disable AN0, AN1, AN2 inputs
        bcf     CM1CON0,C1ON    ;     and comparator 1 -> RB0,RB1 digital
        bcf     CM2CON0,C2ON    ; disable comparator 2 -> RC1 digital

        ; initialise variables
        clrf    digit           ; start with digit = 0

;***** Main loop
main_loop
        ; display digit
        pagesel set7seg
        call    set7seg

        ; delay 1 sec
        DelayMS 1000

        ; increment digit
        incf    digit,f
        movlw   .10
        xorwf   digit,w         ; if digit = 10
        btfsc   STATUS,Z
        clrf    digit           ;   reset it to 0

        ; repeat forever
        pagesel main_loop
        goto    main_loop


;***** LOOKUP TABLES ***************************************************
TABLES  CODE    0x200           ; locate at beginning of a page

; pattern table for 7 segment display on port B
;   RB4 = E, RB1:0 = FG
get7sB  addwf   PCL,f
        retlw   b'010010'       ; 0
        retlw   b'000000'       ; 1
        retlw   b'010001'       ; 2
        retlw   b'000001'       ; 3
        retlw   b'000011'       ; 4
        retlw   b'000011'       ; 5
        retlw   b'010011'       ; 6
        retlw   b'000000'       ; 7
        retlw   b'010011'       ; 8
        retlw   b'000011'       ; 9
```

```
; pattern table for 7 segment display on port C
;   RC4:1 = CDBA
get7sC  addwf   PCL,f
        retlw   b'011110'       ; 0
        retlw   b'010100'       ; 1
        retlw   b'001110'       ; 2
        retlw   b'011110'       ; 3
        retlw   b'010100'       ; 4
        retlw   b'011010'       ; 5
        retlw   b'011010'       ; 6
        retlw   b'010110'       ; 7
        retlw   b'011110'       ; 8
        retlw   b'011110'       ; 9

; Display digit passed in 'digit' variable on 7-segment display
set7seg_R
        movf    digit,w         ; get digit to display
        call    get7sB          ; lookup pattern for port B
        movwf   PORTB           ;   then output it
        movf    digit,w         ; repeat for port C
        call    get7sC
        movwf   PORTC
        retlw   0


        END
```

Multiplexing

To display multiple digits, as in (say) a digital clock, the obvious approach is to extend the method we've just used for a single digit. That is, where one digit requires 7 outputs, two digits would apparently need 14 outputs; four digits would need 28 outputs, etc.

At that rate, you would very quickly run out of output pins, even on the bigger PICs!

A technique commonly used to conserve pins is to *multiplex* a number of displays (and/or inputs – a topic we'll look at another time).

When displays are multiplexed, only one (or a subset) of them is on at any time.

Display multiplexing relies on speed, and human persistence of vision, to create an illusion that a number of displays are on at once, whereas in fact they are being lit rapidly in sequence, so quickly that it appears that they are on continuously.

To multiplex 7-segment displays, it is usual to connect each display in parallel, so that one set of output pins on the PIC is connected to every display, the connections between the modules and to the PIC forming a *bus*.

If the common cathodes were all grounded, every module would display the same digit (feebly, since the output current would be shared between them).

Instead, to allow a different digit to be displayed on each module, the individual displays must be capable of being switched on or off under software control.

For that, transistors are usually used as switches, as illustrated below[2]:



Note that it is not possible to connect the common cathodes directly to the PIC's pins; the combined current from all the segments in a module will be up to 70 mA – too high for a single pin to sink. Instead, the pin is used to switch a transistor on or off.

Almost any low-cost NPN transistor[3], such as a BC547, could be used for this, as is it not a demanding application. It's also possible to use FETs; for example, MOSFETs are usually used to switch high-power devices.

When the output pin is set 'high', the transistor's base is pulled high, turning it 'on'. The 1 kΩ resistors are used to limit the base current to around 4 mA – enough to saturate the transistor, effectively grounding the module's common cathode connection, allowing the display connected to that transistor to light.

These transistors are then used to switch each module on, in sequence, for a short time, while the pattern for that digit is output on the display bus. This is repeated for each digit in the display, quickly enough to avoid visible flicker (preferably at least 70 times per second).

To implement this circuit using the Gooligum baseline training board:

- keep the six shunts in every position of jumper block JP4, connecting segments A-D, F and G to pins RB0-1 and RC1-4
- keep the shunt in position 1 ("RA/RB4") of JP5, connecting segment E to pin RB4
- move the shunt in JP6 to across pins 2 and 3 ("RC5"), connecting digit 1 to the transistor controlled by RC5
- place shunts in jumpers JP8, JP9 and JP10, connecting pins RC5, RB5 and RC0 to their respective transistors

All other shunts should be removed.

---

[2] Again, the diagram reflects the connections on the Gooligum baseline training board, not what looks neatest.

[3] If you had common-anode displays, you would normally use PNP transistors as high-side switches (between VDD and each common anode), instead of the NPN low-side switches shown here.

Example application

To demonstrate display multiplexing, we'll implement a simple timer: the first digit will count minutes and the next two digits will count seconds (00 to 59).

The approach taken in the single-digit example above – set the outputs and then delay for 1 s – won't work, because the display multiplexing has to continue throughout the delay.

Ideally the display multiplexing would be a "background task"; one that continues steadily while the main program is free to perform tasks such as responding to changing inputs. That's an ideal application for timer-based interrupts – a feature available on more advanced PICs (as we will see in midrange lesson 12), but not baseline devices like the 16F506.

But a timer can still be used to good advantage when implementing multiplexing on a baseline PIC. It would be impractical to try to use programmed delays while multiplexing; there's too much going on. But Timer0 can provide a steady *tick* that we can base our timing on – displaying each digit for a single tick, and then counting ticks to decide when a certain time (e.g. 1 sec) has elapsed and we need to perform an action (such as incrementing counters).

If the tick period is too short, there may not be enough time to complete all the program logic needed between ticks, but if it's too long, the display will flicker.

Many PIC developers use a standard 1 ms tick, but to simplify the task of counting in seconds, an (approximately) 2 ms tick is used in this example. If each of three digits is updated at a rate of 2 ms per digit, the whole 3-digit display is updated every 6 ms, so the display rate is 1 ÷ 6 ms = 167 Hz – fast enough to avoid perceptible flicker.

To generate an approximately 2 ms tick, we can use Timer0 in timer mode (based on the 1 MHz instruction clock), with a prescale ratio of 1:256. Bit 2 of Timer0 will then be changing with a period of 2048 µs.

In pseudo-code, the multiplexing technique used here is:

```
time = 0:00
repeat
      tick count = 0
      repeat
            display minutes digit for 1 tick (2 ms)
            display tens digit for 1 tick
            display ones digit for 1 tick
      until tick count = #ticks in 1 second

      increment time by 1 second
forever
```

To store the time, the simplest approach is to use three variables, to store the minutes, tens and ones digits separately. Setting the time to zero then means clearing each of these variables.

To display a single digit, such as minutes, the code becomes:

```
        ; display minutes for 2.048 ms
w60_hi  btfss   TMR0,2          ; wait for TMR0<2> to go high
        goto    w60_hi
        movf    mins,w          ; output minutes digit
        pagesel set7seg
        call    set7seg
        pagesel $
        bsf     MINUTES         ; enable minutes display
w60_lo  btfsc   TMR0,2          ; wait for TMR<2> to go low
        goto    w60_lo
```

This routine begins by waiting for TMR0<2> to go high, then displays the minutes digit (with the others turned off), and finally waits for TMR0<2> to go low again.

The routine to display the tens digit also begins with a wait for TMR0<2> to go high:

```
        ; display tens for 2.048 ms
w10_hi  btfss   TMR0,2          ; wait for TMR0<2> to go high
        goto    w10_hi
        movf    tens,w          ; output tens digit
        pagesel set7seg
        call    set7seg
        pagesel $
        bsf     TENS            ; enable tens display
w10_lo  btfsc   TMR0,2          ; wait for TMR<2> to go low
        goto    w10_lo
```

There is no need to explicitly turn off the minutes digit, since, whenever a new digit pattern is output by the '`set7seg`' routine, the "digit enable" pins, RB5, RC0 and RC5 are always cleared (because the digit pattern tables contain '0's for these bits). Thus, all the displays are blanked whenever a new digit is output.

The ones digit is then displayed in the same way:

```
        ; display ones for 2.048 ms
w1_hi   btfss   TMR0,2          ; wait for TMR0<2> to go high
        goto    w1_hi
        movf    ones,w          ; output ones digit
        pagesel set7seg
        call    set7seg
        pagesel $
        bsf     ONES            ; enable ones display
w1_lo   btfsc   TMR0,2          ; wait for TMR<2> to go low
        goto    w1_lo
```

By waiting for TMR0<2> high at the start of each digit display routine, we can be sure that each digit is displayed for exactly 2.048 ms (or, as close as the internal RC oscillator allows, which is only accurate to 1% or so…).

Note that the '`set7seg`' subroutine has been modified to accept the digit to be displayed as a parameter passed in W, instead of placing it a shared variable; it shortens the code a little to do it this way.

It's also a good idea to blank the display, by clearing the digit enable lines, before outputting the new digit pattern on the display bus – this avoids "ghosting" (visible in low light) due to PORTB being updated while the pattern for the previous digit is still being output on PORTC.

So the '`set7seg`' routine becomes:

```
; Display digit passed in W on 7-segment display
set7seg_R
        ; disable displays
        clrf    PORTB           ; clear all digit enable lines on PORTB
        clrf    PORTC           ;   and PORTC

        ; output digit pattern
        movwf   temp            ; save digit
        call    get7sB          ; lookup pattern for port B
        movwf   PORTB           ;   then output it
        movf    temp,w          ; get digit
        call    get7sC          ;   then repeat for port C
        movwf   PORTC
        retlw   0
```

Note also the '`pagesel $`' after the subroutine call. It is necessary to ensure that the current page is selected before the '`goto`' commands in the wait loops are executed.

After TMR0<2> goes low at the end of the 'ones' display routine, there is approximately 1 ms before it will go high again, when the 'minutes' display will be scheduled to begin again. That means that there is a "spare" 1 ms, after the end of the 'ones' routine, in which to perform the program logic of counting ticks and incrementing the time counters; 1 ms is 1000 instruction cycles – plenty of time!

The following code construct continues multiplexing the digit display until 1 second has elapsed:

```
; multiplex display for 1 sec
        movlw   1000000/2048/3  ; display each of 3 digits for 2.048 ms each
        movwf   mpx_cnt         ;   repeat multiplex loop for 1 second

mplex_loop
        ; display minutes for 2.048 ms

        ; display tens for 2.048 ms

        ; display ones for 2.048 ms

        decfsz  mpx_cnt,f       ; continue to multiplex display
        goto    mplex_loop      ;   until 1 s has elapsed
```

Since there are three digits displayed in the loop, and each is displayed for 2 ms (approx.), the total time through the loop is 6 ms, so the number of iterations until 1 second has elapsed is 1 s ÷ 6 ms = 167, small enough to fit into a single 8-bit counter, which is why a tick period of approximately 2 ms was chosen.

Note that, even if the internal RC oscillator was 100% accurate, giving an instruction clock of exactly 1 MHz, the time taken by this loop will be 162 × 3 × 2.048 ms = 995.3 ms. Hence, this "clock" is guaranteed to be out by at least 0.5%. But accuracy isn't the point of this exercise.

After displaying the current time for (close to) 1 second, we need to increment the time counters, and that can be done as follows:

```
; increment counters
        incf    ones,f          ; increment ones
        movlw   .10
        xorwf   ones,w          ; if ones overflow,
        btfss   STATUS,Z
        goto    end_inc
        clrf    ones            ;   reset ones to 0
        incf    tens,f          ;   and increment tens
        movlw   .6
        xorwf   tens,w          ;   if tens overflow,
        btfss   STATUS,Z
        goto    end_inc
        clrf    tens            ;       reset tens to 0
        incf    mins,f          ;       and increment minutes
        movlw   .10
        xorwf   mins,w          ;       if minutes overflow,
        btfsc   STATUS,Z
        clrf    mins            ;           reset minutes to 0
end_inc
```

It's simply a matter of incrementing the 'ones' digit as was done for a single digit, checking for overflows and incrementing the higher digits accordingly. The overflow (or *carry*) from seconds to minutes is done by testing for "tens = 6". If you wanted to make this purely a seconds counter, counting from 0 to 999 seconds, you'd simply change this to test for "tens = 10", instead.

After incrementing the time counters, the main loop begins again, displaying the updated time.

Complete program

Here is the complete program, incorporating the above code fragments.

One point to note is that TMR0 is never initialised; there's no need, as it simply means that there may be a delay of up to 2 ms before the display begins for the first time, which isn't at all noticeable.

```
;*****************************************************************
;   Description:    Lesson 8, example 2                          *
;                                                                *
;   Demonstrates use of multiplexing to drive multiple 7-seg displays   *
;                                                                *
;   3 digit 7-segment LED display: 1 digit minutes, 2 digit seconds     *
;   counts in seconds 0:00 to 9:59 then repeats,                 *
;   with timing derived from int 4 MHz oscillator                *
;                                                                *
;*****************************************************************
;   Pin assignments:                                             *
;       RB0-1,RB4,RC1-4 = 7-segment display bus (common cathode) *
;       RC5             = minutes enable (active high)           *
;       RB5             = tens enable                            *
;       RC0             = ones enable                            *
;                                                                *
;*****************************************************************

    list        p=16F506
    #include    <p16F506.inc>

    radix       dec


;***** CONFIGURATION
                ; ext reset, no code protect, no watchdog, 4 MHz int clock
    __CONFIG    _MCLRE_ON & _CP_OFF & _WDT_OFF & _IOSCFS_OFF & _IntRC_OSC_RB4EN

; pin assignments
    #define MINUTES PORTC,5     ; minutes enable
    #define TENS    PORTB,5     ; tens enable
    #define ONES    PORTC,0     ; ones enable


;***** VARIABLE DEFINITIONS
        UDATA_SHR
temp    res 1                   ; used by set7seg routine (temp digit store)

        UDATA
mpx_cnt res 1                   ; multiplex counter
mins    res 1                   ; current count: minutes
tens    res 1                   ;   tens
ones    res 1                   ;   ones


;***** RC CALIBRATION
RCCAL   CODE    0x3FF           ; processor reset vector
        res 1                   ; holds internal RC cal value, as a movlw k


;***** RESET VECTOR ****************************************************
RESET   CODE    0x000           ; effective reset vector
        movwf   OSCCAL          ; apply internal RC factory calibration
        pagesel start
        goto    start           ; jump to main code
```

```
;***** Subroutine vectors
set7seg                         ; display digit on 7-segment display
        pagesel set7seg_R
        goto    set7seg_R


;***** MAIN PROGRAM ****************************************************
MAIN    CODE

;***** Initialisation
start
        ; configure ports
        clrw                    ; configure PORTB and PORTC as all outputs
        tris    PORTB
        tris    PORTC
        clrf    ADCON0          ; disable AN0, AN1, AN2 inputs
        bcf     CM1CON0,C1ON    ;     and comparator 1 -> RB0,RB1 digital
        bcf     CM2CON0,C2ON    ; disable comparator 2 -> RC0,RC1 digital

        ; configure timer
        movlw   b'11010111'     ; configure Timer0:
                ; --0-----          timer mode (T0CS = 0) -> RC5 usable
                ; ----0---          prescaler assigned to Timer0 (PSA = 0)
                ; -----111          prescale = 256 (PS = 111)
        option                  ;   -> increment every 256 us
                                ;      (TMR0<2> cycles every 2.048ms)
        ; initialise variables
        banksel mins            ; start with count = 0:00
        clrf    mins
        clrf    tens
        clrf    ones

;***** Main loop
main_loop

; multiplex display for 1 sec
        movlw   1000000/2048/3  ; display each of 3 digits for 2.048 ms each
        movwf   mpx_cnt         ;   repeat multiplex loop for approx 1 second

mplex_loop
        ; display minutes for 2.048 ms
w60_hi  btfss   TMR0,2          ; wait for TMR0<2> to go high
        goto    w60_hi
        movf    mins,w          ; output minutes digit
        pagesel set7seg
        call    set7seg
        pagesel $
        bsf     MINUTES         ; enable minutes display
w60_lo  btfsc   TMR0,2          ; wait for TMR<2> to go low
        goto    w60_lo

        ; display tens for 2.048 ms
w10_hi  btfss   TMR0,2          ; wait for TMR0<2> to go high
        goto    w10_hi
        movf    tens,w          ; output tens digit
        pagesel set7seg
        call    set7seg
        pagesel $
        bsf     TENS            ; enable tens display
w10_lo  btfsc   TMR0,2          ; wait for TMR<2> to go low
        goto    w10_lo
```

```
        ; display ones for 2.048 ms
w1_hi   btfss   TMR0,2          ; wait for TMR0<2> to go high
        goto    w1_hi
        movf    ones,w          ; output ones digit
        pagesel set7seg
        call    set7seg
        pagesel $
        bsf     ONES            ; enable ones display
w1_lo   btfsc   TMR0,2          ; wait for TMR<2> to go low
        goto    w1_lo

        decfsz  mpx_cnt,f       ; continue to multiplex display
        goto    mplex_loop      ;   until 1 sec has elapsed

; increment time counters
        incf    ones,f          ; increment ones
        movlw   .10
        xorwf   ones,w          ; if ones overflow,
        btfss   STATUS,Z
        goto    end_inc
        clrf    ones            ;   reset ones to 0
        incf    tens,f          ;   and increment tens
        movlw   .6
        xorwf   tens,w          ;   if tens overflow,
        btfss   STATUS,Z
        goto    end_inc
        clrf    tens            ;       reset tens to 0
        incf    mins,f          ;       and increment minutes
        movlw   .10
        xorwf   mins,w          ;       if minutes overflow,
        btfsc   STATUS,Z
        clrf    mins            ;           reset minutes to 0
end_inc

; repeat forever
        goto    main_loop


;***** LOOKUP TABLES ************************************************
TABLES  CODE    0x200           ; locate at beginning of a page

; pattern table for 7 segment display on port B
;   RB4 = E, RB1:0 = FG
get7sB  addwf   PCL,f
        retlw   b'010010'       ; 0
        retlw   b'000000'       ; 1
        retlw   b'010001'       ; 2
        retlw   b'000001'       ; 3
        retlw   b'000011'       ; 4
        retlw   b'000011'       ; 5
        retlw   b'010011'       ; 6
        retlw   b'000000'       ; 7
        retlw   b'010011'       ; 8
        retlw   b'000011'       ; 9

; pattern table for 7 segment display on port C
;   RC4:1 = CDBA
get7sC  addwf   PCL,f
        retlw   b'011110'       ; 0
        retlw   b'010100'       ; 1
        retlw   b'001110'       ; 2
```

```
        retlw   b'011110'       ; 3
        retlw   b'010100'       ; 4
        retlw   b'011010'       ; 5
        retlw   b'011010'       ; 6
        retlw   b'010110'       ; 7
        retlw   b'011110'       ; 8
        retlw   b'011110'       ; 9

; Display digit passed in W on 7-segment display
set7seg_R
        ; disable displays
        clrf    PORTB           ; clear all digit enable lines on PORTB
        clrf    PORTC           ;   and PORTC

        ; output digit pattern
        movwf   temp            ; save digit
        call    get7sB          ; lookup pattern for port B
        movwf   PORTB           ;   then output it
        movf    temp,w          ; get digit
        call    get7sC          ;   then repeat for port C
        movwf   PORTC
        retlw   0


        END
```

Binary-Coded Decimal

In the previous example, each digit in the time count was stored in its own 8-bit variable.

Since a single digit can only have values from 0 to 9, while an 8-bit register can store any integer from 0 to 255, it is apparent that storing each digit in a separate variable is an inefficient use of storage space. That can be an issue on devices with such a small amount of data memory – only 67 bytes on the 16F506.

The most space-efficient way to store integers is to use pure binary representation. E.g. the number '183' would be stored in a single byte as b'10110111' (or 0xB7). That's three digits in a single byte. Of course, 3-digit numbers larger than 255 need two bytes, but any 4-digit number can be stored in two bytes, as can any 5-digit number less than 65536.

The problem with such "efficient" binary representation is that it's difficult (i.e. time consuming) to unpack into decimal; necessary so that it can be displayed.

Consider how you would convert a number such as 0xB7 into decimal.

First, determine how many hundreds are in it. Baseline PICs do not have a "divide" instruction; the simplest approach is to subtract 100, check to see if there is a borrow, and subtract 100 again if there wasn't (keeping track of the number of hundreds subtracted; this number of hundreds is the first digit):

$$0xB7 - 100 = 0x53$$

Now continue to subtract 10 from the remainder (0x53) until a borrow occurs, keeping track of how many tens were successfully subtracted, giving the second digit:

$$0x53 - (8 \times 10) = 0x03$$

The remainder (0x03) is of course the third digit.

Not only is this a complex routine, and takes a significant time to run (up to 12 subtractions are needed for a single conversion), it also requires storage; intermediate results such as "remainder" and "tens count" need to be stored somewhere.

Sometimes converting from pure binary into decimal is unavoidable, perhaps for example when dealing with quantities resulting from an analog to digital conversion (which we'll look at in lesson 10). But often, when storing numbers which will be displayed in decimal form, it makes sense to store them using *binary-coded decimal* representation.

In binary-coded decimal, or *BCD*, two digits are *packed* into each byte – one in each nybble (or "nibble", as Microchip spells it).

For example, the BCD representation of 56 is 0x56. That is, each decimal digit corresponds directly to a hex digit when converted to BCD.

All eight bits in the byte are used, although not as efficiently as for binary. But BCD is far easier to work with for decimal operations, as we'll see.

Example application

To demonstrate the use of BCD, we'll modify the previous example to store "seconds" as a BCD variable.

So only two variables for the time count are now needed, instead of three:

```
        UDATA
mpx_cnt res 1                   ; multiplex counter
mins    res 1                   ; time count: minutes
secs    res 1                   ;   seconds (BCD)
```

To display minutes is the same as before (since minutes is still being stored in its own variable), but to display the tens digit, we must first extract the digit from the high nybble, as follows:

```
        ; display tens for 2.048 ms
w10_hi  btfss   TMR0,2          ; wait for TMR0<2> to go high
        goto    w10_hi
        swapf   secs,w          ; get tens digit
        andlw   0x0F            ;   from high nybble of seconds
        pagesel set7seg
        call    set7seg         ;   then output it
        pagesel $
```

To move the contents of bits 4-7 (the high nybble) into bits 0-3 (the low nybble) of a register, you could use four '`rrf`' instructions, to shift the contents of the register four bits to the right.

But the baseline PICs provide a very useful instruction for working with BCD: '`swapf f,d`' – "**swap** nybbles in **f**ile register". As usual, '`f`' is the register supplying the value to be swapped, and '`d`' is the destination: '`,f`' to write the swapped result back to the register, or '`,w`' to place the result in W.

Having gotten the tens digit into the lower nybble (in W, since we don't want to change the contents of the '`secs`' variable), the upper nybble has to be cleared, so that only the tens digit is passed to the '`set7seg`' routine.

This is done through a technique called *masking*.

It relies on the fact that any bit ANDed with '1' remains unchanged, while any bit ANDed with '0' is cleared to '0'. That is:

n AND 1 = n

n AND 0 = 0

So if a byte is ANDed with binary 00001111, the high nybble will be cleared, leaving the low nybble unchanged.

So far we've only seen the exclusive-or instructions, but the baseline PICs provide equivalent instructions for the logical "and" and "or" operations, including '`andlw`', which ANDs a literal value with the contents of W, placing the result in W – "**and l**iteral with **W**".

So the '`andlw 0x0F`' instruction masks off the high nybble, leaving only the tens digit left in W, to be passed to the '`set7seg`' routine.

And why express the *bit mask* in hexadecimal (0x0F) instead of binary (b'00001111')? Simply because, when working with BCD values, hexadecimal notation seems clearer.

Extracting the ones digit is simply a masking operation, as the ones digit is already in the lower nybble:

```
        ; display ones for 2.048 ms
w1_hi   btfss   TMR0,2          ; wait for TMR0<2> to go high
        goto    w1_hi
        movf    secs,w          ; get ones digit
        andlw   0x0F            ;   from low nybble of seconds
        pagesel set7seg
        call    set7seg         ;   then output it
        pagesel $
```

The only other routine that has to be done differently, due to storing seconds in BCD format, is incrementing the time count, as follows:

```
; increment time counters
        incf    secs,f          ; increment seconds
        movf    secs,w          ; if ones overflow,
        andlw   0x0F
        xorlw   .10
        btfss   STATUS,Z
        goto    end_inc
        movlw   .6              ;   BCD adjust seconds
        addwf   secs,f
        movlw   0x60
        xorwf   secs,w          ;   if seconds = 60,
        btfss   STATUS,Z
        goto    end_inc
        clrf    secs            ;       reset seconds to 0
        incf    mins,f          ;       and increment minutes
        movlw   .10
        xorwf   mins,w          ;       if minutes overflow,
        btfsc   STATUS,Z
        clrf    mins            ;           reset minutes to 0
end_inc
```

To check to see whether the 'ones' digit has been incremented past 9, it is extracted (by masking) and tested to see if it equals 10. If it does, then we need to reset the 'ones' digit to 0, and increment the 'tens' digit. But remember that BCD digits are essentially hexadecimal digits. The 'tens' digit is really counting by 16s, as far as the PIC is concerned, which operates purely on binary numbers, regardless of whether we consider them to be in BCD format. If the 'ones' digit is equal to 10, then adding 6 to it would take it to 16, which would overflow, leaving 'ones' cleared to 0, and incrementing 'tens'.

Putting it another way, you could say that adding 6 adjusts for BCD digit overflow. Some microprocessors provide a "decimal adjust" instruction, that performs this adjustment. The PIC doesn't, so we do it manually.

Finally, note that to check for seconds overflow, the test is not for "seconds = 60", but "seconds = 0x60", i.e. the value to be compared is expressed in hexadecimal, because seconds is stored in BCD format. Forgetting to express the seconds overflow test in hex would be an easy mistake to make…

The rest of the code is exactly the same as before, so won't be repeated here (although the source files for all the examples are of course available for download from www.gooligum.com.au).

Overall, the BCD version uses 104 words of program memory, and 4 bytes of data memory, compared with 102 words and 5 bytes for the un-packed version.

Is saving a single byte of data memory worth the additional complexity and two extra words of program memory? In this example, probably not. But in cases where you need to store more data, adding instructions to pack and extract that data can be well worth while. As with any trade-off, "it depends".

Conclusion

That completes our survey of digital I/O with the baseline PIC devices. More is possible, of course, but to go much further in digital I/O, it is better to make the jump to the midrange architecture.

But before doing so, we'll take a look at analog inputs, using comparators (in lesson 9) and analog to digital conversion (in lesson 10).